
Examination Guidelines 
 

We will soon come to the end of the formal instruction for this semester.  We hope that you enjoyed 

this module and that you have enjoyed the learning experience. 

The exam paper has a total of 100 marks.  The duration of the paper is 2 hours. It is a complete fill in 

examination paper. 

Please note that the examination paper is in English only. You are welcome to answer the questions in 

English or Afrikaans or both.  

IMPORTANT: YOU WILL NOT BE ALLOWED TO USE A CALCULATOR IN THE 

EXAMINATION FOR THIS MODULE! 

Question 1 (30 marks) consists of multiple-choice questions on both the theory (Stallings) and the 

practical work.  

Some of the other questions are discussion questions on the theory covered in the COS2621 syllabus. 

Use the theory questions contained in the assignments and the self-tests, as well as the questions 

included in past examination papers as guidelines for the kinds of question you might expect in this 

section. There are also questions on the use of the DEBUG utility.  In the last section you will be 

required to write program code in assembly language. The examination paper will include an appendix 

listing assembler instructions. You are expected to provide sufficient comments in your programs so 

that the logic will be easy to follow. 

If you completed both assignments, understand the content as outlined in tutorial letters 101, 102, 103, 

201, 202, and the discussion of the self-tests, and you have worked through the assembler program 

examples using both DEBUG and Nasm, you should find the paper relatively easy.  

When you need to do calculations in a question, you should show how you have arrived at the answer.  

If you only write down the answer without showing any steps and you made an arithmetic error 

somewhere, it is not possible to award any marks for an incorrect answer.  If you show the steps, you 

will get credit for correct reasoning. For question 1, if calculations are required, you only need to write 

down the alternative you choose. 

If you have to answer a question where you have to compare two techniques, for example, it is not 

sufficient to state that one technique is better than the other.  You have to substantiate your answer with 

a reason. 

Answer essay-type questions in telegram style.  Do not waste time writing down irrelevant facts.  From 

our side, we will try to put the questions as clearly as possible so that you will know exactly what is 

expected with each question. 

Assignment 2 solution 
 

Question 1            [20] 

a) Why is it important for a Computer Scientist to study computer organisation? 

Refer to pages 7 for the definition of computer organisation: ”Computer organization refers to the 

operational units and their interconnections that realize the architectural specifications. Examples of 

architectural attributes include the instruction set, the number of bits used to represent various data 

types (e.g., numbers, characters), I/O mechanisms, and techniques for addressing memory. 

Organizational attributes include those hardware details transparent to the programmer, such as control 

signals; interfaces between the computer and peripherals; and the memory technology used.”  

 

The reason why it is important is examined in the preface on page XIV: “It has always been important to 

design computer systems to achieve high performance, but never has this requirement been stronger or 

more difficult to satisfy than today. All of the basic performance characteristics of computer systems, 

including processor speed, memory speed, memory capacity, and interconnection data rates, are 

increasing rapidly. Moreover, they are increasing at different rates. This makes it difficult to design a 

balanced system that maximizes the performance and utilization of all elements. Thus, computer design 

increasingly becomes a game of changing the structure or function in one area to compensate for a 

performance mismatch in another area. We will see this game played out in numerous design decisions 

throughout the book” 

 

 

b) Explain the following terms in relation to microprocessor speed: 

(i) Branch prediction 

(ii) Data flow analysis 

(iii) Speculative execution 

(iv) Pipelining 

 

Refer to page 38: 

Pipelining: With pipelining, a processor can simultaneously work on multiple instructions. The processor 

overlaps operations by moving data or instructions into a conceptual pipe with all stages of the pipe 

processing simultaneously. For example, while one instruction is being executed, the computer is 

decoding the next instruction. 

 

Branch prediction: The processor looks ahead in the instruction code fetched from memory and predicts 

which branches, or groups of instructions, are likely to be processed next. If the processor guesses right 

most of the time, it can prefetch the correct instructions and buffer them so that the processor is kept 

busy. The more sophisticated examples of this strategy predict not just the next branch but multiple 

branches ahead. Thus, branch prediction increases the amount of work available for the processor to 

execute. 

Data flow analysis: The processor analyzes which instructions are dependent on each other’s results, or 

data, to create an optimized schedule of instructions. In fact, instructions are scheduled to be executed 

when ready, independent of the original program order. This prevents unnecessary delay. 

 

Speculative execution: Using branch prediction and data flow analysis, some processors speculatively 

execute instructions ahead of their actual appearance in the program execution, holding the results in 

temporary locations. This enables the processor to keep its execution engines 

 

Question 2  

Assembly language programming:  

Problem statement: Write a program in IBM PC assembly language to do the following: 1. Display the 
following prompt: "Enter the number for the factorial calculation (0 to 8):" 2 Accept the number from the 
keyboard as an ASCII string. 3. Convert the string to a numeric value. 4. Call a subroutine that calculates 
the factorial of the number. 5. Convert the result to an ASCII string and display it on the screen. 6. 
Terminate the program when a key such as "Q" is pressed. N.B. Keep the following in mind: 1. The value 
must be passed to both subroutines as a parameter on the stack. The result should also be passed back 
to the calling program on the stack. 2. Use either of the following two algorithms as guidelines for your 
subroutines: 

Iterative function FactIter (N): 
Result = 1; 
For I = 2 to N do 
Result = Result * I; 
FactIter = Result; 
Recursive function FactRec (N): 
If N = 1 then 
FactRec = 1 
else 
FactRec = N * FactRec (N-1); 
 
You can find the .lst and the .asm file under additional materials. 
 
At the end we also present a more comprehensive solution from a previous year. 

bits 16 

org 0x100 

; 

; Factorial.asm 

; Program to convert input number 0 to 8 to its factorial 

; 

jmp main 

mess1: db 'Enter number for factorial calculation (0 to 8): ',0dh,0ah,'$' 

mess2: db 0dh,0ah,'The factorial is: ',0dh,0ah,'$' 

inbuff: db 02h,0,'  ' 

outbuff: db '    ',0dh,0ah,'$' 

; 

; Display an ASCII string on the screen 

; Terminate with $ 

display: 

mov ah,09 ; display message on the screen 

int 21h 

ret 

; 

; Accept an ASCII string from the keyboard 

; Stored from 3rd byte on 

getnum: 

mov dx,inbuff ; accept ASCII string from keyboard 

mov ah,0ah 

int 21h 

ret 

; 

; Convert an ASCII string to a numerical value 

; 

convert: 

xor ax,ax ; initial value of AX = 0 

xor bh,bh ; BH = 0 

mov cx,10 ; to build integer in AX (multiply by 10) 

mov si,dx ; DX points to start of input buffer 

next_char: 

mov bl,[si] ; move contents of memory pointed to by SI to BL. 

cmp bl,0Dh ; is it a carriage return? 

je done ; yes, we are done 

cmp bl,38h ; ASCII for the character '8' is 38h 

jg error ; > '8', invalid character 

sub bl,30h ; convert to numeric value (ASCII '0' = 30h) 

jl error ; < 0, invalid character 

imul cx ; DX:AX = AX * 10 

add ax,bx ; add next digit 

inc si ; pointer to next char 

jmp next_char ; repeat for next character 

error: 

mov al,'E' ; flag an error 

done: 

ret  ; return to calling program result returnd in AX 

; 

; Calculate factorial 

; 

fact: 

pop bx ; store return address temporarily 

pop ax ; value for calculation 

mov cx,ax ; CX is used as a loop counter 

dec cx   

repeat: 

imul cx ; DX:AX = AX*CX (note only one operand is specified) 

  ; We have two 16 bit operands and a 32 bit result 

loop repeat ; Decrement CX. Repeat loop if CX<>0 

push ax ; result on stack 

push bx ; restore calling address 

ret 

; 

; Convert a numerical value to an ASCII string 

; 

conv2: 

mov di,outbuff ; address of output buffer 

add di,4 ; point to end of output buffer 

mov bx,10 ; we divide by 10 

loop1: 

xor dx,dx ; set DX to 0 for the divide operation 

idiv bx ; divide DX:AX by BX, 

  ; result in AX and remainder in DX 

add dx,30h ; convert remainder to ASCII 

mov [di],dl ; move digit into output string (one byte) 

cmp ax,0 ; is result = 0? 

je endconv ; yes, then we have finished 

dec di ; pointer one back 

jmp loop1 ; repeat loop 

endconv: 

ret 

; 

; Main program 

; 

main: 

mov dx,mess1 

call display ; display prompt 

call getnum ; get number from keyboard 

mov dx,inbuff+2 ; address of input string 

call convert ; convert to numeric value 

cmp al,'E' ; is the number within range? 

je nok 

cmp al,00h    ; By convention 0! = 1 

jne notzero 

inc al 

notzero: 

cmp ax,03h    ; If AX = 1 or 2 nothing to do 

jb ok      

push ax ; parameter is passed on the stack 

call fact ; calculate factorial 

pop ax ; result returned on stack 

ok: 

call conv2 ; convert result to ASCII string 

mov dx,mess2 ; display 'result' message 

call display 

mov dx,outbuff 

call display ; display result 

nok:  ; not ok 

int 20h ; terminate program 

 

 
     1                                  bits 16 
     2                                  org 0x100 

     3                                  ; 

     4                                  ; Factorial.asm 

     5                                  ; Program to convert input number 0 to 8 to its factorial 

     6                                  ; 

     7 00000000 E9B200                  jmp main 

     8 00000003 456E746572206E756D-     mess1: db 'Enter number for factorial calculation (0 to 8): ',0dh,0ah,'$' 

     9 0000000C 62657220666F722066- 

    10 00000015 6163746F7269616C20- 

    11 0000001E 63616C63756C617469- 

    12 00000027 6F6E20283020746F20- 

    13 00000030 38293A200D0A24      

    14 00000037 0D0A54686520666163-     mess2: db 0dh,0ah,'The factorial is: ',0dh,0ah,'$' 

    15 00000040 746F7269616C206973- 

    16 00000049 3A200D0A24          

    17 0000004E 02002020                inbuff: db 02h,0,'  ' 

    18 00000052 202020200D0A24          outbuff: db '    ',0dh,0ah,'$' 

    19                                  ; 

    20                                  ; Display an ASCII string on the screen 

    21                                  ; Terminate with $ 

    22                                  display: 

    23 00000059 B409                    mov ah,09 ; display message on the screen 

    24 0000005B CD21                    int 21h 

    25 0000005D C3                      ret 

    26                                  ; 

    27                                  ; Accept an ASCII string from the keyboard 

    28                                  ; Stored from 3rd byte on 

    29                                  getnum: 

    30 0000005E BA[4E00]                mov dx,inbuff ; accept ASCII string from keyboard 

    31 00000061 B40A                    mov ah,0ah 

    32 00000063 CD21                    int 21h 

    33 00000065 C3                      ret 

    34                                  ; 

    35                                  ; Convert an ASCII string to a numerical value 

    36                                  ; 

    37                                  convert: 

    38 00000066 31C0                    xor ax,ax ; initial value of AX = 0 

    39 00000068 30FF                    xor bh,bh ; BH = 0 

    40 0000006A B90A00                  mov cx,10 ; to build integer in AX (multiply by 10) 

    41 0000006D 89D6                    mov si,dx ; DX points to start of input buffer 

    42                                  next_char: 

    43 0000006F 8A1C                    mov bl,[si] ; move contents of memory pointed to by SI to BL. 

    44 00000071 80FB0D                  cmp bl,0Dh ; is it a carriage return? 

    45 00000074 7414                    je done  ; yes, we are done 

    46 00000076 80FB38                  cmp bl,38h ; ASCII for the character '8' is 38h 

    47 00000079 7F0D                    jg error ; > '8', invalid character 

    48 0000007B 80EB30                  sub bl,30h ; convert to numeric value (ASCII '0' = 30h) 

    49 0000007E 7C08                    jl error ; < 0, invalid character 

    50 00000080 F7E9                    imul cx  ; DX:AX = AX * 10 

    51 00000082 01D8                    add ax,bx ; add next digit 

    52 00000084 46                      inc si  ; pointer to next char 

    53 00000085 E9E7FF                  jmp next_char ; repeat for next character 

    54                                  error: 

    55 00000088 B045                    mov al,'E' ; flag an error 

    56                                  done: 

    57 0000008A C3                      ret  ; return to calling program result returnd in AX 

    58                                  ; 

    59                                  ; Calculate factorial 

    60                                  ; 

    61                                  fact: 

    62 0000008B 5B                      pop bx  ; store return address temporarily 

    63 0000008C 58                      pop ax  ; value for calculation 

    64 0000008D 89C1                    mov cx,ax ; CX is used as a loop counter 

    65 0000008F 49                      dec cx   

    66                                  repeat: 

    67 00000090 F7E9                    imul cx  ; DX:AX = AX*CX (note only one operand is specified) 

    68                                    ; We have two 16 bit operands and a 32 bit result 

    69 00000092 E2FC                    loop repeat ; Decrement CX. Repeat loop if CX<>0 

    70 00000094 50                      push ax  ; result on stack 

    71 00000095 53                      push bx  ; restore calling address 

    72 00000096 C3                      ret 

    73                                  ; 

    74                                  ; Convert a numerical value to an ASCII string 

    75                                  ; 

    76                                  conv2: 

    77 00000097 BF[5200]                mov di,outbuff ; address of output buffer 

    78 0000009A 81C70400                add di,4 ; point to end of output buffer 

    79 0000009E BB0A00                  mov bx,10 ; we divide by 10 

    80                                  loop1: 

    81 000000A1 31D2                    xor dx,dx ; set DX to 0 for the divide operation 

    82 000000A3 F7FB                    idiv bx  ; divide DX:AX by BX, 

    83                                    ; result in AX and remainder in DX 

    84 000000A5 81C23000                add dx,30h ; convert remainder to ASCII 

    85 000000A9 8815                    mov [di],dl ; move digit into output string (one byte) 

    86 000000AB 3D0000                  cmp ax,0 ; is result = 0? 

    87 000000AE 7404                    je endconv ; yes, then we have finished 

    88 000000B0 4F                      dec di  ; pointer one back 

    89 000000B1 E9EDFF                  jmp loop1 ; repeat loop 

    90                                  endconv: 

    91 000000B4 C3                      ret 

    92                                  ; 

    93                                  ; Main program 

    94                                  ; 

    95                                  main: 

    96 000000B5 BA[0300]                mov dx,mess1 

    97 000000B8 E89EFF                  call display ; display prompt 

    98 000000BB E8A0FF                  call getnum ; get number from keyboard 

    99 000000BE BA[5000]                mov dx,inbuff+2 ; address of input string 

   100 000000C1 E8A2FF                  call convert ; convert to numeric value 

   101 000000C4 3C45                    cmp al,'E' ; is the number within range? 

   102 000000C6 741F                    je nok 

   103 000000C8 3C00                    cmp al,00h      ; By convention 0! = 1 

   104 000000CA 7502                    jne notzero 

   105 000000CC FEC0                    inc al 

   106                                  notzero: 

   107 000000CE 3D0300                  cmp ax,03h      ; If AX = 1 or 2 nothing to do 

   108 000000D1 7205                    jb ok      

   109 000000D3 50                      push ax  ; parameter is passed on the stack 

   110 000000D4 E8B4FF                  call fact ; calculate factorial 

   111 000000D7 58                      pop ax  ; result returned on stack 

   112                                  ok: 

   113 000000D8 E8BCFF                  call conv2 ; convert result to ASCII string 

   114 000000DB BA[3700]                mov dx,mess2 ; display 'result' message 

   115 000000DE E878FF                  call display 

   116 000000E1 BA[5200]                mov dx,outbuff 

   117 000000E4 E872FF                  call display ; display result 

   118                                  nok:  ; not ok 

   119 000000E7 CD20                    int 20h  ; terminate program 

   120                                   

   121                                   

   122        

 
Another example using a DEBUG unassembly 
 
     a 100   

426F:0100  ;   

426F:0100  ; FACT.COM   

426F:0100  ;   

426F:0100  ; This program calculates the factorial of a number that   

426F:0100  ; is input from the keyboard. First the factorial is calculated   

426F:0100  ; using an iterative subroutine and then the calculation is done   

426F:0100  ; using a recursive subroutine. The result is displayed on the   

426F:0100  ; screen in both cases.   

426F:0100  ;   

426F:0100     jmp    300          ; Branch to main program   

426F:0103  ;   

426F:0103  ; Data area   

426F:0103  ;   

426F:0103  ; Parameter buffer for input   

426F:0103  ; First byte gives the maximum length of the string including the  

426F:0103  ; carriage return   

426F:0103  ; Second byte will contain the actual length of the string   

426F:0103  ; after the interrupt   

426F:0103  ; Input buffer starts at third byte   

426F:0103     db     4   

426F:0104     db     0   

426F:0105     db     '1234'   

426F:0109  ; Prompt for input   

426F:0109     db  'Please enter the number that you want the factorial of.'   

426F:0140     db     ' (Maximum 3 digits.)',0d,0a,'$'   

426F:0157  ; First result   

426F:0157     db     'Iterative result: '   

426F:0169     db     '     ',0d,0a,'$'   

426F:0171  ; Second result   

426F:0171     db     'Recursive result: '   

426F:0183     db     '     ',0d,0a,'$'   

426F:018B    

 a 1a0   

426F:01A0  ;   

426F:01A0  ; Clear_Screen   

426F:01A0  ;   

426F:01A0  ; PRECONDITION:  None   

426F:01A0  ; POSTCONDITION: The screen is cleared   

426F:01A0  ;   

426F:01A0     mov    ax,0600      ; Scroll the entire screen.   

426F:01A3                         ; (Refer to Study Guide p. 109)   

426F:01A3     mov    bh,07        ; Colour: white on black   

426F:01A5     mov    cx,0000      ; From top left corner   

426F:01A8     mov    dx,184f      ; To lower right corner   

426F:01AB     int    10           ; BIOS interrupt   

426F:01AD     ret   

426F:01AE    

 a 1b0   

426F:01B0  ;   

426F:01B0  ; Display_Message   

426F:01B0  ;   

426F:01B0  ; PRECONDITION:  DX contains the address of a character string   

426F:01B0  

  ;                terminated by a '$'.   

426F:01B0  ; POSTCONDITION: The string is displayed on the screen.   

426F:01B0  ;   

426F:01B0     mov    ah,09        ; Display message   

426F:01B2     int    21           ; DOS interrupt   

426F:01B4     ret   

426F:01B5    

 a 1c0   

426F:01C0  ;   

426F:01C0  ; Accept_String   

426F:01C0  ;   

426F:01C0  ; PRECONDITION:  DX contains the address of the parameter buffer.  

426F:01C0  

  ; POSTCONDITION: A string is accepted from the keyboard and 

426F:01C0  ;                stored in the buffer specified.   

426F:01C0  ;   

426F:01C0     mov    ah,0a        ; Accept a string from the keyboard   

426F:01C2     int    21           ; DOS interrupt   

426F:01C4     ret   

426F:01C5    

 a 1d0   

426F:01D0  ;   

426F:01D0  ; Set_Cursor   

426F:01D0  ;   

426F:01D0  ; PRECONDITION:  DH contains the row (0..24).   

426F:01D0  ;                DL contains the column (0..79).   

426F:01D0  ; POSTCONDITION: The cursor is set at the specified row:column.   

426F:01D0 ;   

426F:01D0     mov    ah,02        ; Set cursor   

426F:01D2     mov    bh,00        ; Screen number 0   

426F:01D4     int    21           ; DOS interrupt   

426F:01D6     ret   

426F:01D7    

 a 200   

426F:0200  ;   

426F:0200  ; Convert_To_Numeric   

426F:0200  ;   

426F:0200  ; PRECONDITION:  SI contains the address of a character string.   

426F:0200  ;                The string contains characters in the range  

426F:0200  ;                '0..9'.   

426F:0200  ;                A carriage return indicates end of the string.   

426F:0200  ;                Length of string <= 3.   

426F:0200  ; POSTCONDITION: The numeric equivalent of the ASCII character   

426F:0200  ;                string is returned in AX.   

426F:0200  ;   

426F:0200     mov    cx,3         ; Loop count. Maximum 3 characters.   

426F:0203     mov    bl,0a        ; 0Ah = 10 decimal   

426F:0205     xor    ax,ax        ; AX = 0   

426F:0207     xor    dx,dx        ; DX = 0   

426F:0209     mov    dl,[si]      ; Get next character   

426F:020B     cmp    dl,0d        ; Is it a CR?   

426F:020E     je     218          ; Yes, end of string   

426F:0210     imul   dl           ; AX = AX * 10   

426F:0212     sub    dl,30        ; Get numeric value of character   

426F:0215                         ; ASCII 30h = '0'   

426F:0215     add    ax,dx        ; Add new digit   

426F:0217     inc    si           ; Adjust character pointer   

426F:0218     loop   209          ; Loop if there are more   

426F:021A     ret   

426F:021B    

 a 220   

426F:0220 ;   

426F:0220 ; Convert_To_ASCII   

426F:0220 ;   

426F:0220 ; PRECONDITION:  AX contains the value to be converted.   

426F:0220 ;                DI contains the address of the buffer   

426F:0220 ;                where the resulting string must be stored.   

426F:0220 ; POSTCONDITION: The ASCII equivalent of the value in AX is  

426F:0220 ;                returned in the buffer.   

426F:0220 ;   

426F:0220     push   bx           ; Store value of BX   

426F:0221     mov    cx,5         ; Maximum 5 characters   

426F:0224     mov    dx,0         ; DX = 0 for divide   

426F:0227     mov    bx,0a        ; BL = 10   

426F:022A     add    di,4         ; Point to end of buffer   

426F:022D     idiv   bx           ; AX = quotient   

426F:022F                         ; DX = remainder   

426F:022F     add    dx,30        ; Convert digit to ASCII character   

426F:0232     mov    [di],dl      ; Store in output string   

426F:0234     dec    di           ; Adjust output pointer   

426F:0235     cmp    ax,0         ; Have we finished?   

426F:0238     je     23F          ; Yes, return   

426F:023A     mov    dx,0         ; Clear DX   

426F:023D     loop   22D          ; Next division   

426F:023F     pop    bx           ; Restore BX   

426F:0240     ret   

426F:0241    

 a 250   

426F:0250 ;   

426F:0250 ; Iterative_Factorial   

426F:0250 ;   

426F:0250 ; PRECONDITION:  The number of which the factorial must be  

426F:0250 ;                calculated (N), is on top of the stack   

426F:0250 ;                (two byte value).   

426F:0250 ;                Result is unpredictable if N <= 1 or N > 9.   

426F:0250 ; POSTCONDITION: The result is returned on the stack   

426F:0250 ;                (two byte value).   

426F:0250 ;   

426F:0250     pop    di           ; Store return address   

426F:0251     pop    ax           ; Get hold of value N.   

426F:0252     mov    cx,ax        ;   

426F:0254     dec    cx           ; CX = N 1 (Loop N 1 times)   

426F:0255     imul   cx           ; AX = AX * N i   

426F:0257     loop   255          ; Decrement CX, loop if CX <> 0   

426F:0259     push   ax           ; Store result on the stack   

426F:025A     push   di           ; Restore return address   

426F:025B     ret   

426F:025C    

 a 270   

426F:0270 ;   

426F:0270 ; Recursive_Factorial   

426F:0270 ;   

426F:0270 ; PRECONDITION:  The number of which the factorial must be 426F:0270 ;                calculated 

is on the stack (two byte value).  

426F:0270 ;                On top of the stack, space is reserved  

426F:0270 ;                for the two byte result.   

426F:0270 ;                The result is unpredictable if N <= 1 or N > 9.   

426F:0270 ; POSTCONDITION: The result is returned on the stack   

426F:0270 ;                (two byte value).   

426F:0270 ;   

426F:0270     mov    bp,sp        ; Set up frame pointer   

426F:0272     add    bp,4         ; Point to parameter   

426F:0275     mov    bx,[bp]      ; Load calling parameter   

426F:0278     cmp    bx,1         ; Is it the base case?   

426F:027B     je     297          ; Yes, initial calculation   

426F:027D     sub    bx,1         ; AX = AX   1   

426F:0280     push   bx           ; Store AX for the next call   

426F:0281     push   bx           ; Reserve space for result   

426F:0282     call   270          ; Recursive call with n 1   

426F:0285     pop    ax           ; Accumulated total   

426F:0286     pop    bx           ; Get rid of previous parameter   

426F:0287     mov    bp,sp        ; Adjust stack frame   

426F:0289     add    bp,4   

426F:028C     mov    bx,[bp]      ; Next multiplicant   

426F:028F     imul   bx           ; AX = AX * BX   

426F:0291     sub    bp,2   

426F:0294     mov    [bp],ax      ; Store accumulated result   

426F:0297     ret   

426F:0298    

 a 300   

426F:0300 ;   

426F:0300 ; Main_Program   

426F:0300 ;   

426F:0300     call   1a0          ; Clear the screen   

426F:0303     mov    dh,0b        ; Set cursor at: line number 12   

426F:0305     mov    dl,00        ;                column number 0   

426F:0307     call   1d0          ; Set the cursor   

426F:030A     mov    dx,109       ; Address of prompt   

426F:030D     call   1b0          ; Display prompt   

426F:0310     mov    dx,103       ; Address of parameter buffer   

426F:0313     call   1c0          ; Accept number from keyboard 

426F:0313                         ; (ASCII string)   

426F:0316     mov    si,105       ; Address of input string   

426F:0319     call   200          ; Convert string to numeric   

426F:031C     cmp    ax,9         ; Is number > 9?   

426F:031F     jg     349          ; Yes, print error message   

426F:0321     mov    bx,ax        ; Store result temporarily   

426F:0323     push   ax           ; Number to top of stack   

426F:0324 ; Iterative calculation   

426F:0324     call   250          ; Calculate factorial iteratively   

426F:0327     pop    ax           ; Get result from stack   

426F:0328     mov    di,169       ; Give address of output buffer   

426F:032B     call   220          ; Convert result to ASCII string   

426F:032E     mov    dx,157       ; Address of first result display   

426F:0331     call   1b0          ; Display iterative result   

426F:0334 ; Recursive calculation   

426F:0334     push   bx           ; Number to top of stack   

426F:0335     push   bx           ; Reserve space for result   

426F:0336     call   270          ; Calculate factorial recursively   

426F:0339     pop    ax           ; Get result from stack   

426F:033A     pop    bx           ; Get rid of parameter   

426F:033B     mov    di,183       ; Give address of output buffer   

426F:033E     call   220          ; Convert to ASCII   

426F:0341     mov    dx,171       ; Address of second result display   

426F:0344     call   1b0          ; Display recursive result   

426F:0347     int    20           ; Terminate program   

426F:0349 ; Input error   

426F:0349     mov    dx,360       ; Address of error message   

426F:034C     call   1b0          ; Display message   

426F:034F     jmp    30a          ; Get a new number   

426F:0351    

 a 360   

426F:0360     db     'Input number too large',0a,0d,'$'   

426F:0379    

 n fact.com   

 r bx   

BX 0000  

:0000   

 r cx   

CX 0000  

:300   

 w   

Writing 00300 bytes  

 q                          

